Instructions for Converting POLYMATH Solutions to Excel Worksheets - Introduction

WHY EXCEL FOR NUMERICAL PROBLEM SOLVING?

SPREADSHEETS ARE THE COMPUTATIONAL TOOLS MOST WIDELY USED BY CHEMICAL ENGINEERS. PROVIDING THE CAPABILITY FOR NUMERICAL PROBLEM SOLVING EXTENDS CONSIDERABLY THE COMPUTATIONAL POTENTIAL OF THE ENGINEER

WHY USE A POLYMATH PREPROCESSOR ?

THE MATHEMATICAL MODEL CAN BE MUCH EASIER AND FASTER CODED AND DEBUGGED USING POLYMATH. THE POLYMATH MODEL SERVES AS BASIS FOR THE SPREADSHEET MODEL WHERE THE VARIABLE NAMES ARE REPLACED BY THEIR ADDRESSES. IT ALSO SERVES AS AN EASY TO UNDERSTAND DOCUMENTATION OF THE MODEL

CONVERTING POLYMATH SOLUTIONS TO EXCEL WORKSHEETS

TYPES OF PROBLEMS DISCUSSED

1 ONE NONLINEAR ALGEBRAIC EQUATION – GOAL SEEK – SLIDES 3-9

2 SYSTEMS OF NONLINEAR ALGEBRAIC EQUATIONS – SOLVER – SLIDES 10-14

3 ODE – INITIAL VALUE PROBLEMS – 4^{TH} ORDER EXPLICIT RK – SLIDES 15-24

4 ODE – BOUNDARY VALUE – EXPLICIT EULER + GOAL SEEK – SLIDES 25-29

5 DAE – INITIAL VALUE PROBLEMS – IMPLICIT EULER – SLIDES 30-36

6 PDE – INITIAL VALUE - METHOD OF LINES – EXPLICIT EULER – SLIDES 37-40

7 MULTIPLE LINEAR REGRESSION – LINEST – SLIDES 41-45

8 POLYNOMIAL REGRESSION – LINEST – SLIDES 46-49

9 MULTIPLE NONLINEAR REGRESSION – SOLVER – SLIDES 50-53

One Nonlinear Algebraic Equation Instructions for Conversion (1)

To obtain a basic solution of a system containing one implicit nonlinear algebraic equation and several explicit equations the **POLYMATH equations** should be converted to **Excel formulas** and then the **"Goal Seek" tool** can be used. In order to obtain a well documented Excel worksheet, which can be easily modified for parametric runs it is recommended to carry out the conversion in the following steps:

1. Copy **the implicit equation** and the ordered explicit equations from the POLYMATH solution report.

2. Paste the equations into an Excel worksheet; remove the text and the equation numbers.

3. Rearrange the equations in the order: **constant definitions**, **functions of the constants**, **parameter definitions**, **unknown**, **explicit functions of the unknown** and **implicit function** of the unknown.

One Nonlinear Algebraic Equation

Instructions for Conversion (2)

4. Copy the right hand side of the equations into the adjacent cell and replace the variable names by variable addresses. Note that "If" statements and some functions may require additional rewriting and/or rearrangement. Use **absolute addressing for the constants** and the functions of constant and **relative addressing for the unknown and its functions** (Note that pressing F4 converts selected reference from relative to absolute). In the cell adjacent to the **unknown put its initial estimate.**

5. Use the **"Goal Seek"** tool to set the value of the cell containing the implicit function of the unknown at zero while changing the value in the cell of the unknown..

One Nonlinear Algebraic Equation

Ordered POLYMATH File

The use of this procedure is demonstrated in reference to Demo 2.

Nonlinear equations

[1] f(V) = (P+a/(V^2))*(V-b)-R*T = 0

Explicit equations

[1] P = 56
[2] R = 0.08206
[3] T = 450
[4] Tc = 405.5
[5] Pc = 111.3
[6] Pr = P/Pc
[7] a = 27*(R^2*Tc^2/Pc)/64
[8] b = R*Tc/(8*Pc)
[9] Z = P*V/(R*T)

One Nonlinear Algebraic Equation

Excel Formulas

	A	B	C
3			Equations
4			
5	Constants	P = 56	=56
6		R = 0.08206	=0.08206
7		T = 450	=450
8		Tc = 405.5	=405.5
9		Pc = 111.3	=111.3
10	Functions of the constants	Pr = P/Pc	=$C$5/$C$9
11		a = 27*(R^2*Tc^2/Pc)/64	=27*($C$6^2*$C$8^2/$C$9)/64
12		b = R*Tc/(8*Pc)	=$C$6*$C$8/(8*$C$9)
13	Unknown	V	0.7
14	Functions of the unknown	Z = P*V/(R*T)	=$C$5*C13/($C$6*$C$7)
15		f(V) = (P+a/(V^2))*(V-b)-R*T = 0	=($C$5+$C$11/(C13^2))*(C13-$C$12)-$C$6*$C$7

One Nonlinear Algebraic Equation Solution

To solve the nonlinear equation in **cell C15 "Goal Seek"** is used to **set the value** in this cell **at zero** while **changing** the contents of cell **C13**.

3		**Initial values**	**Solution**
4			
5	P = 56	56	56
6	R = 0.08206	0.08206	0.08206
7	T = 450	450	450
8	Tc = 405.5	405.5	405.5
9	Pc = 111.3	111.3	111.3
10	Pr = P/Pc	0.50314	0.50314
11	a = 27*(R^2*Tc^2/Pc)/64	4.19695	4.19695
12	b = R*Tc/(8*Pc)	0.03737	0.03737
13	V	**0.7**	**0.57489**
14	Z = P*V/(R*T)	1.06155	0.87183
15	f(V) = (P+a/(V^2))*(V-b)-R*T = 0	**5.85576**	**8.4999E-07**

One Nonlinear Algebraic Equation

Modifying the Equation Set

The example is next solved for **Pr = 1, 2, 4, 10 and 20**. To achieve this, the parameter Tr and its function P=Pr*Pc are added to the equation set and the cells containing the unknown and its functions are copied and modified as necessary.

	A	B	C
23	Parameter	Pr	1
24	Function of the parameter	P = Pr*Pc	=C23*$C$9
25	Unknown	V	0.233508696752435
26	Functions of the unknown	Z = P*V/(R*T)	=C24*C25/($C$6*$C$7)
27		f(V) = (P+a/(V^2))*(V-b)-R*T = 0	=(C24+$C$11/(C25^2))*(C25-$C$12)-$C$6*$C$7

One Nonlinear Algebraic Equation

Complete solution set

To obtain the solution for other values of Pr cells 24 – 27 of column C are copied and the value of Pr entered in row 23. "**Goal Seek**" is applied separately to **every column** containing a different Pr value.

Pr	1	2	4	10	20
P = Pr*Pc	111.3	222.6	445.2	1113	2226
V	**0.23351**	**0.07727**	**0.06065**	**0.05088**	**0.04618**
Z = P*V/(R*T)	0.70381	0.46578	0.73126	1.53341	2.78348
f(V) = (P+a/(V^2))*(V-b)-R*T = 0	**3.940E-06**	**7.604E-07**	**2.208E-06**	**6.184E-08**	**6.962E-09**

Systems of Nonlinear Algebraic Equations

Instructions for Conversion (1)

To obtain a basic solution of a system containing several implicit nonlinear algebraic equations the **POLYMATH equations** are converted to **Excel formulas** and then the **"Solver" tool** is used. The recommended steps for conversion are:

1. Copy the **implicit equations** and the **ordered explicit equations** from the POLYMATH solution report.

2. Paste the equations into an Excel worksheet; remove the text and the equation numbers.

3. Rearrange the equations in the order: **constant definitions**, **functions of the constants**, **parameter definitions**, **unknowns**, **explicit functions of the unknowns** and **implicit functions of the unknowns.**

4. Add an equation with the **sum of squares of the implicit functions**.

Systems of Nonlinear Algebraic Equations

Instructions for Conversion (2)

5. Copy the right hand side of the equations into the adjacent cell and replace the variable names by variable addresses. Use **absolute addressing for the constants** and the functions of constant and **relative addressing for the unknowns** and functions of the unknowns. In the cell adjacent to the unknowns put initial estimates.

6. Use the **"Solver" tool** to set the value of the cell containing the **sum of squares** of the implicit functions of the unknowns at zero (or **minimizing its value**) while changing the values in the cells of the unknowns.

Systems of Nonlinear Algebraic Equations

Ordered POLYMATH File

The use of this procedure is demonstrated in reference to Demo 5

Nonlinear equations

[1] f(CD) = CC*CD-KC1*CA*CB = 0
[2] f(CX) = CX*CY-KC2*CB*CC = 0
[3] f(CZ) = CZ-KC3*CA*CX = 0

Explicit equations

[1] KC1 = 1.06
[2] CY = CX+CZ
[3] KC2 = 2.63
[4] KC3 = 5
[5] CA0 = 1.5
[6] CB0 = 1.5
[7] CC = CD-CY
[8] CA = CA0-CD-CZ
[9] CB = CB0-CD-CY

Systems of Nonlinear Algebraic Equations Excel Formulas

	A	B	C
4			**Equations**
5	Constants	CA0 = 1.5	1.5
6		CB0 = 1.5	1.5
7		KC1 = 1.06	1.06
8		KC2 = 2.63	2.63
9		KC3 = 5	5
10	Unknowns	CD	0
11		CX	0
12		CZ	0
13	Functions of the unknowns	CY = CX+CZ	=C11+C12
14		CC = CD-CY	=C10-C13
15		CA = CA0-CD-CZ	=$C$5-C10-C12
16		CB = CB0-CD-CY	=$C$6-C10-C13
17		f(CD) = CC*CD-KC1*CA*CB = 0	=C14*C10-$C$7*C15*C16
18		f(CX) = CX*CY-KC2*CB*CC = 0	=C11*C13-$C$8*C16*C14
19		f(CZ) = CZ-KC3*CA*CX = 0	=C12-$C$9*C15*C11
20	Sum of squares of errors	sum=f(CD)^2+f(CX)^2+f(CZ)^2	=C17^2+C18^2+C19^2

Systems of Nonlinear Algebraic Equations Excel Formulas

The **"Solver" tool** is used to **minimize the sum of squares of errors in cell C20** by setting C20 as "target cell" and searching for its minimal value by **changing cells C10, C11 and C12**.

	Initial values	Solution
CA0 = 1.5	1.5	1.5
CB0 = 1.5	1.5	1.5
KC1 = 1.06	1.06	1.06
KC2 = 2.63	2.63	2.63
KC3 = 5	5	5
CD	**0**	**0.70533**
CX	**0**	**0.17779**
CZ	**0**	**0.37398**
CY = CX+CZ	0	0.55177
CC = CD-CY	0	0.15357
CA = CA0-CD-CZ	1.5	0.42069
CB = CB0-CD-CY	1.5	0.24290
f(CD) = CC*CD-KC1*CA*CB = 0	-2.385	5.1760E-09
f(CX) = CX*CY-KC2*CB*CC = 0	0	-1.3358E-07
f(CZ) = CZ-KC3*CA*CX = 0	0	-2.9923E-07
sum=f(CD)^2+f(CX)^2+f(CZ)^2	**5.68823**	**1.0741E-13**

ODE – Initial Value Problems

The Runge-Kutta Method

There are no tools in Excel to solve differential equations so the solution algorithm must be build into the solution worksheet. In this example a **fixed step size, explicit, fourth-order Runge-Kutta** algorithm is used. The system of *N* first-order ODE for the functions

$y_i, \quad i = 1, \ldots, N$ is written :

$$\frac{dy_i(x)}{dx} = f_i(x, y_1, \ldots, y_N), \qquad i = 1, \ldots, N \tag{1}$$

The fourth-order Runge-Kutta formula is written:

$$k_1 = hf(x_n, y_n)$$

$$k_2 = hf(x_n + \frac{h}{2}, y_n + \frac{k_1}{2}) \tag{2}$$

$$k_3 = hf(x_n + \frac{h}{2}, y_n + \frac{k_2}{2})$$

$$k_4 = hf(x_n + h, y_n + k_3)$$

$$y_{n+1} = y_n + \frac{1}{6}(k_1 + 2k_2 + 2k_3 + k_4)$$

This formula advances a solution from x_n to $x_{n+1} \equiv x_n + h$

ODE – Initial Value Problems Instructions for Conversion (1)

Apply the Runge-Kutta algorithm to the system of first-order, ODE carry out the conversion from the POLYMATH file to the Excel spreadsheet in the following steps:

1.Copy the differential equation and the ordered explicit algebraic equations from the POLYMATH solution report.

2. Paste the equations into an Excel worksheet; remove the text and the equation numbers.

3. Put the parameters: **final value of the independent variable** and **integration step-size (h)** in the first cells of the worksheet. Rearrange the equations in the order: **constant definitions**, **functions of the constants**, **independent variable**, **dependent variables**, **explicit functions of the variables** and **differential equations**.

ODE – Initial Value Problems

Instructions for Conversion (2)

4. Copy the right hand side of the equations into the adjacent cell and replace the variable names by variable addresses. Use **absolute addressing for the constants** and the functions of constant and **relative addressing for the variables** and functions of the variables. In the cell adjacent to the **variables put their initial values**.

5. Copy the section starting with the independent variable up to the end of the equation set and paste this section three times below, to **obtain the values of k_2, k_3 and k_4**. Change the equations as needed to reflect the change in the variable values, as shown in Equation (2).

6. In the next column write the equations to calculate the advanced values of the independent and dependent variables.

7. **Copy and paste the columns** (or rows) as many time as needed in order **to reach the final value of the independent variable.**

ODE – Initial Value Problems

Ordered POLYMATH File

The use of this procedure is demonstrated in reference to Demo 9.

Differential equations as entered by the user

[1] d(T1)/d(t) = (W*Cp*(T0-T1)+UA*(Tsteam-T1))/(M*Cp)

[2] d(T2)/d(t) = (W*Cp*(T1-T2)+UA*(Tsteam-T2))/(M*Cp)

[3] d(T3)/d(t) = (W*Cp*(T2-T3)+UA*(Tsteam-T3))/(M*Cp)

Explicit equations as entered by the user

[1] W = 100

[2] Cp = 2.0

[3] T0 = 20

[4] UA = 10.

[5] Tsteam = 250

[6] M = 1000

ODE – Initial Value Problems Excel Formulas (1)

	A	B	C
4		**Definitions**	**Equations/values**
5	Final value (ind. Var.)	tf=200	200
6	Integration step size	h	=($C$5-$C$13)/200
7	Constants	W=100	=100
8		Cp=2.0	=2
9		T0=20	=20
10		UA=10.	=10
11		Tsteam=250	=250
12		M=1000	=1000
13	Independent variable	t	0
14	Dependent variables	T1	20
15		T2	20
16		T3	20
17	Differential equations	k11=h*d(T1)/d(t)=h*(W*Cp*(T0-T1)+UA*(Tsteam-T1))/(M*Cp)	=$C$6*($C$7*$C$8*($C$9-C14)+$C$10*($C$11-C14))/($C$12*$C$8)
18		k12=h*d(T2)/d(t)=h*(W*Cp*(T1-T2)+UA*(Tsteam-T2))/(M*Cp)	=$C$6*($C$7*$C$8*(C14-C15)+$C$10*($C$11-C15))/($C$12*$C$8)

ODE – Initial Value Problems

Excel Formulas (2)

	A	B	C
20		T1+k11/2	=C14+$C$6*C17/2
21		T2+k12/2	=C15+$C$6*C18/2
22		T3+k13/2	=C16+$C$6*C19/2
23		k21	=$C$6*($C$7*$C$8*($C$9-C20)+$C$10*($C$11-C20))/($C$12*$C$8)
24		k22	=$C$6*($C$7*$C$8*(C20-C21)+$C$10*($C$11-C21))/($C$12*$C$8)
25		k23	=$C$6*($C$7*$C$8*(C21-C22)+$C$10*($C$11 C22))/($C$12*$C$8)
26		T1+k21/2	=C14+$C$6*C23/2
27		T2+k22/2	=C15+$C$6*C24/2
28		T3+k23/2	=C16+$C$6*C25/2
29		k31	=$C$6*($C$7*$C$8*($C$9-C26)+$C$10*($C$11-C26))/($C$12*$C$8)
30		k32	=$C$6*($C$7*$C$8*(C26-C27)+$C$10*($C$11-C27))/($C$12*$C$8)
31		k33	=$C$6*($C$7*$C$8*(C27-C28)+$C$10*($C$11-C28))/($C$12*$C$8)

ODE – Initial Value Problems

Excel Formulas (3)

	A	B	C
32		T1+k31	=C14+$C$6*C29
33		T2+k32	=C15+$C$6*C30
34		T3+k33	=C16+$C$6*C31
35		k41	=$C$6*($C$7*$C$8*($C$9-C32)+$C$10*($C$11-C32))/($C$12*$C$8)
36		k42	=$C$6*($C$7*$C$8*(C32-C33)+$C$10*($C$11-C33))/($C$12*$C$8)
37		k43	=$C$6*($C$7*$C$8*(C33-C34)+$C$10*($C$11-C34))/($C$12*$C$8)

In column D the **solution is advanced from** x_n to $x_{n+1} \equiv x_n + h$

	A	B	C	D
13	Independent variable	t	0	=C13+$C$6
14	Dependent variables	T1	20	=C14+(1/6)*(C17+2*C23+2*C29+C35)
15		T2	20	=C15+(1/6)*(C18+2*C24+2*C30+C36)
16		T3	20	=C16+(1/6)*(C19+2*C25+2*C31+C37)

ODE – Initial Value Problems

Results for t=1 min and t=80 min (1)

	Definitions	Equations/values		
Final value (ind. var.)	tf=200	200		
Integration step size	h	1		
Constants	W=100	100		
	Cp=2.0	2		
	T0=20	20		
	UA=10.	10		
	Tsteam=250	250		
	M=1000	1000		
Independent variable	t	**0**	**1**	**80**
Dependent variables	T1	**20**	**21.09168**	**30.94992**
	T2	**20**	**21.14532**	**41.35871**
	T3	**20**	**21.14708**	**51.19303**
Differential equations	k11	1.15	1.03537	2.5860E-04
	k12	1.15	1.13891	2.3274E-03
	k13	1.15	1.14409	1.0603E-02

ODE – Initial Value Problems

Results for t=1 min and t=80 min (2)

	T1+k11/2	20.575	21.60937	30.95005
	T2+k12/2	20.575	21.71477	41.35987
	T3+k13/2	20.575	21.71913	51.19833
	k21	1.089625	0.98102	2.4502E-04
	k22	1.147125	1.13089	2.2181E-03
	k23	1.147125	1.14097	1.0162E-02
	T1+k21/2	20.54481	21.58219	30.95004
	T2+k22/2	20.57356	21.71076	41.35982
	T3+k23/2	20.57356	21.71757	51.19811
	k31	1.09279	0.98387	2.4574E-04
	k32	1.14426	1.12859	2.2232E-03
	k33	1.14713	1.14073	1.0180E-02
	T1+k31	21.09279	22.07555	30.95016
	T2+k32	21.14426	22.27391	41.36093
	T3+k33	21.14713	22.28781	51.20321
	k41	1.03526	0.93207	2.3280E-04
	k42	1.13913	1.11880	2.1185E-03
	k43	1.14398	1.13717	9.7559E-03

ODE – Initial Value Problems

Plot of the results



ODE – Boundary Value Problems

Solution Method

There are no tools in Excel to solve differential equations so the solution algorithm must be build into the solution worksheet. In this example a **fixed step size, explicit, Euler algorithm** is used. After setting up the worksheet for integrating the differential equations the **"Goal Seek"** (for the case of one boundary value) or the **"Solver"** (for the case of several boundary values) is used for converging to the proper initial values.

The formula for the Euler method is

$$y_{n+1} = y_n + hf(x_n, y_n) \tag{3}$$

This formula advances a solution from x_n to $x_{n+1} \equiv x_n + h$

Steps of the Solution.

1. Copy the **differential equation** and the **ordered explicit algebraic equations** from the POLYMATH solution report.

2. Paste the equations into an Excel worksheet; remove the text and the equation numbers.

ODE – Boundary Value Problems

Steps of the Solution

3. Put the parameters: **final value of the independent variable** and **integration step-size (h)** in the first cells of the worksheet. Rearrange the equations in the order: **constant definitions**, **functions of the constants**, **independent variable**, **dependent variables**, **explicit functions** of the variables and **differential equations**.

4. Copy the right hand side of the equations into the adjacent cell and replace the variable names by variable addresses. In the cell adjacent to the variables put their initial values. If the initial value is not known put initial estimates, instead.

5. In the next column write the equations to calculate the **advanced values of the variables** using Equation 3.

6. Copy and paste the columns as many times as needed in order to reach the final value of the independent variable.

7. Use the "**Goal Seek**" (for the case of **one boundary value**) or the "**Solver**" (for the case of **several boundary values**) to converge to the **desired final value** of the variables while **changing their initial values**.

ODE – Boundary Value Problems

POLYMATH File and Excel Formulas

The use of this procedure is demonstrated in reference to Demo 8.

Differential equations as entered by the user

[1] d(CA)/d(z) = y

[2] d(y)/d(z) = k*CA/DAB

Explicit equations as entered by the user

[1] k = 0.001

[2] DAB = 1.2E-9

	A	B	C	D
4		**Definitions**	**Equations/values**	
5				
6	Final value ind.var.	zf=0.001	0.001	
7	Integration step-size	h	=(C6-C10)/100	
8	Constants	k = 0.001	0.001	
9		DAB = 1.2E-9	0.0000000012	
10	Independent variable	z	0	**=C10+$C$7**
11	Dependent variables	CA	0.2	**=C11+$C$7*C13**
12		y	-150	**=C12+$C$7*C14**
13	Differential equations	f1=d(CA)/d(z) = y	=C12	=D12
14		f2=d(y)/d(z) = k*CA/DAB	=$C$8*C11/$C$9	=$C$8*D11/$C$9

ODE – Boundary Value Problems
Results at z = 0, 0.00001 and 0.001 m

1. **Initial estimate: y = -150**

Independent variable	z	0	0.00001	0.001
Dependent variables	CA	0.2	0.1985	0.11708
	y	**-150**	-148.333	**-26.052**
Differential equations	f1=d(CA)/d(z) = y	-150	-148.333	-26.052
	f2=d(y)/d(z) = k*CA/DAB	1.6667E+05	165416.667	9.7563E+04

2. **After using of the "Goal Seek"** tool to set the value of y(0.001) at zero while changing y(0).

Independent variable	z	0	0.00001	0.001
Dependent variables	CA	0.2	0.001	0.13770
	y	**-131.913**	0.19868	**-7.53E-14**
Differential equations	f1=d(CA)/d(z) = y	-131.913	0.13770	-7.53E-14
	f2=d(y)/d(z) = k*CA/DAB	1.6667E+05	-130.247	1.1475E+05

DAE – Initial Value Problems

Solution Method

There are no tools in Excel to solve differential equations so the solution algorithm must be build into the solution worksheet. In this example a fixed step size, **implicit, Euler algorithm** is used. Using this method the differential equations are converted into nonlinear algebraic equations. Thus, in each integration step a system of nonlinear algebraic equations is solved using the **"Solver"** tool. The formula for the implicit Euler method is

$$F_n = y_n - \left[y_{n-1} + \frac{h}{2}\{ f(x_{n-1}, y_{n-1}) + f(x_n, y_n) \} \right] = 0 \qquad (4)$$

This formula advances a solution from x_{n-1} to $x_n \equiv x_{n-1} + h$ for $n>1$.

DAE – Initial Value Problems

Steps of the Solution (1)

1. Copy the **differential equations** and the **ordered explicit algebraic equations** from the POLYMATH solution report.

2. Paste the equations into an Excel worksheet; remove the text and the equation numbers.

3. Put the parameters: **final value of the independent variable** and **integration step-size (h)** in the first cells of the worksheet. Rearrange the equations in the order: **constant definitions**, **functions of the constants**, **independent variable**, **dependent variables**, **explicit functions** of the variables, **differential equations** and **implicit algebraic equations**.

4. **Add an equation with the sum of squares of the implicit functions** (the algebraic equations and the implicit Euler method representation of the differential equations).

DAE – Initial Value Problems

Steps of the Solution (2)

5. Copy the right hand side of the equations into the adjacent cells and replace the variable names by variable addresses. Use absolute addressing for the constants and the functions of constant and relative addressing for the variables and functions of the variables. In the cell adjacent to the variables put their initial values. In the cell containing the sum of squares of the function values include only the functions associated with the implicit algebraic equations.

6. **Use the "Solver" (or "Goal Seek" tools) to find the initial values** of the unknowns associated with the implicit algebraic equations.

7. In the next column write the equations to calculate the advanced values of the independent and dependent variables.

8. From this point on the columns can be copied and pasted, as many time as needed to reach the final value of the independent variable. The **"Solver" tool must be applied on the columns sequentially**, to solve the system of nonlinear algebraic equations for each step

DAE – Initial Value Problems

POLYMATH File

The use of this procedure is demonstrated in reference to Demo 11 The differential equations and the ordered explicit algebraic equations as copied from the POLYMATH solution report are the following.

Differential equations as entered by the user

```
[1] d(L)/d(x2) = L/(k2*x2-x2)
[2] d(T)/d(x2) = Kc*err
```

Explicit equations as entered by the user

```
[1] Kc = 0.5e6
[2] k2 = 10^(6.95464-1344.8/(T+219.482))/(760*1.2)
[3] x1 = 1-x2
[4] k1 = 10^(6.90565-1211.033/(T+220.79))/(760*1.2)
[5] err = (1-k1*x1-k2*x2)
```

DAE – Initial Value Problems Excel formulas

	A	B	C
5		**Definitions**	**Equations/values**
6	Final value ind.var.	x2(f)=	0.8
7	Integration step-size	h	=($C$6-$C$8)/20
8	Independent variable	x2	0.4
9	Dependent variables	L=	100
10		T=	95
11	Explicit equations	x1 = 1-x2	= 1-C8
12		k1 = 10^(6.90565-1211.033/(T+220.79))/(760*1.2)	= 10^(6.90565-1211.033/(C10+220.79))/(760*1.2)
13		k2 = 10^(6.95464-1344.8/(T+219.482))/(760*1.2)	= 10^(6.95464-1344.8/(C10+219.482))/(760*1.2)
14	Differential equations	f1=d(L)/d(x2) = L/(k2*x2-x2)	= C9/(C13*C8-C8)
15		f2=f(T)=(1-k1*x1-k2*x2)=0	=(1-C12*C11-C13*C8)
16	Sum of squares of errors	$[L_n-(L_{n-1}+h/2(f_{1n}+f_{1n-1}))]$^2+$f_{2n+1}$^2	

In the next column (column D) the definition of the independent variable is changed to: **=C8+$C$7** and the definition of the sum of squares of errors is changed to: **=(D9-(C9+($C$7/2)*(C14+D14)))^2+D15^2** .

DAE – Initial Value Problems

Results for x2 = 0.4 and 0.42

Results obtained by applying **"Goal Seek" to set cell C15 at zero while changing** the initial temperature (cell **C10)** and subsequently applying the **"Solver" tool to minimize the value in cell D16** while **changing** the contents of cells **D9 and D10.**

Final value ind.var.	x2(f)=	0.8	
Integration step-size	h	0.02	
Independent variable	x2	0.4	0.42
Dependent variables	L=	100	**89.976**
	T=	**95.583**	**96.142**
Explicit equations	x1 = 1-x2	0.6	0.58
	k1 = 10^(6.90565-1211.033/(T+220.79))/(760*1.2)	1.3116	1.3321
	k2 = 10^(6.95464-1344.8/(T+219.482))/(760*1.2)	0.53250	0.54185
Differential equations	f1=d(L)/d(x2) = L/(k2*x2-x2)	-534.754	-467.601
	f2=f(T)=(1-k1*x1-k2*x2)=0	**6.8122E-05**	-2.1100E-04
Sum of squares of errors	$[L_n-(L_{n-1}+h/2(f_{1n}+f_{1n-1}))]^2+f_{2n+1}^2$		**4.4521E-08**

DAE – Initial Value Problems

Results for x2 = 0.8

Column D is copied and pasted as many time as necessary to reach the final value of x_2 (= 0.8). The "Solver" tool is applied sequentially, for every column to minimize the value in row 16.

Independent variable	x2	**0.8**
Dependent variables	L=	**14.006**
	T=	**108.595**
Explicit equations	x1 = 1-x2	0.2
	k1 =	1.8579
	k2 =	0.78634
Differential equations	f1=d(L)/d(x2) = L/(k2*x2-x2)	3.1544E-04
	f2=f(T)=(1-k1*x1-k2*x2)=0	-6.4625E-04
Sum of squares of errors	$[L_n-(L_{n-1}+h/2(f_{1n}+f_{1n-1}))]^2+f_{2n+1}^2$	**5.1713E-07**

Partial Differential Equations Excel Formulas for Demo 12 (1)

The system of PDEs is converted into a system of first order ODEs using the **method of lines**. Explicit Euler's method is used for solution.

	A	B	C	D
6		**Definitions**	**Equations/values**	
7	Final value ind.var.	tf=6000	6000	
8	Integration Step-size	h	=(C7-C13)/200	
9	Constants	T1 = 0	0	
10		alpha = 2.e-5	0.00002	
11		deltax = .10	0.1	
12		alpha/deltax^2	=C10/C11^2	
13	Independent Variable	t	0	=C13+$C$8
14	Variables	T2	100	=C14+($F$29-$F$28)*C24
15		T3	100	=C15+($F$29-$F$28)*C25
16		T4	100	=C16+($F$29-$F$28)*C26
17		T5	100	=C17+($F$29-$F$28)*C27
18		T6	100	=C18+($F$29-$F$28)*C28
19		T7	100	=C19+($F$29-$F$28)*C29
20		T8	100	=C20+($F$29-$F$28)*C30
21		T9	100	=C21+($F$29-$F$28)*C31
22		T10	100	=C22+($F$29-$F$28)*C32
23		T11 = (4*T10-T9)/3	=(4*C$22-C$21)/3	=(4*D$22-D$21)/3

Partial Differential Equations

Excel Formulas for Demo 12 (2)

	A	B	C
24	Differential equations	f1=d(T2)/d(t) = alpha/deltax^2*(T3-2*T2+T1)	=$C$12*($H28-2*$G28+$C$9)
25		f2=d(T3)/d(t) = alpha/deltax^2*(T4-2*T3+T2)	=$C$12*($I28-2*$H28+$G28)
26		f3=d(T4)/d(t) = alpha/deltax^2*(T5-2*T4+T3)	=$C$12*($J28-2*$I28+$H28)
27		f4=d(T5)/d(t) = alpha/deltax^2*(T6-2*T5+T4)	=$C$12*($K28-2*$J28+$I28)
28		f5=d(T6)/d(t) = alpha/deltax^2*(T7-2*T6+T5)	=$C$12*($L28-2*$K28+$J28)
29		f6=d(T7)/d(t) = alpha/deltax^2*(T8-2*T7+T6)	=$C$12*($M28-2*$L28+$K28)
30		f7=d(T8)/d(t) = alpha/deltax^2*(T9-2*T8+T7)	=$C$12*($N28-2*$M28+$L28)
31		f8=d(T9)/d(t) = alpha/deltax^2*(T10-2*T9+T8)	=$C$12*($O28-2*$N28+$M28)
32		f9=d(T10)/d(t) = alpha/deltax^2*(T11-2*T10+T9)	=$C$12*($P28-2*$O28+$N28)

Column D for this section is obtained by copying and pasting the same section in column C. To obtain the complete solution column D is copied and pasted as many times as needed for reaching the final time.

Partial Differential Equations

Results for t=0, 30 and 6000 min

Definitions	Equations/values		
tf=6000	6000		
h	30		
T1 = 0	0		
alpha = 2.e-5	2.00E-05		
deltax = .10	0.1		
alpha/deltax^2	2.00E-03		
t	**0**	**30**	**6000**
T2	**100**	**94**	**16.1632**
T3	100	100	31.6589
T4	100	100	45.8980
T5	100	100	58.4307
T6	100	100	68.9780
T7	100	100	77.4300
T8	100	100	83.8125
T9	100	100	88.2343
T10	100	100	90.8270
T11 = (4*T10-T9)/3	100	100	91.6912

Partial Differential Equations

Plot of Some Results for Demo 12



Multiple Linear Regression

Copying the Data from POLYMATH

In this demonstration Riedel's equation is fitted to the data of Demo 6.



	TK	logP1	Trec	logT	T2	logP
01	236.45	0	0.00422922	2.3737393	55908.603	0
02	253.55	0.69897	0.003944	2.4040636	64287.603	0.69897
03	261.65	1.	0.0038219	2.4177207	68460.723	1.
04	270.55	1.30103	0.00369617	2.4322475	73197.303	1.30103
05	280.75	1.60206	0.00356189	2.4483198	78820.563	1.60206
06	288.55	1.7781513	0.0034656	2.4602211	83261.103	1.7781513
07	299.25	2.	0.00334169	2.4760342	89550.563	2.
08	315.35	2.30103	0.00317108	2.4987928	99445.623	2.30103
09	333.75	2.60206	0.00299625	2.5234213	1.1139E+05	2.60206
10	353.25	2.8808136	0.00283086	2.5480822	1.2479E+05	2.8808136
11						
12						
13						

Multiple Linear Regression

Pasting the Data into Excel and Adding Titles

	A	B	C	D
1	**Trec**	**logT**	**T2**	**logP**
2	0.00422922	2.3737393	55908.6	0
3	0.003944	2.4040636	64287.6	0.69897
4	0.0038219	2.4177207	68460.72	1
5	0.00369617	2.4322475	73197.3	1.30103
6	0.00356189	2.4483198	78820.56	1.60206
7	0.0034656	2.4602211	83261.1	1.778151
8	0.00334169	2.4760342	89550.56	2
9	0.00317108	2.4987928	99445.62	2.30103
10	0.00299625	2.5234213	1.11E+05	2.60206
11	0.00283086	2.5480822	1.25E+05	2.880814

Multiple Linear Regression

Using the LINEST Function

The LINEST function puts the full set of results in an **area that includes 5 rows and number of columns as the number of the parameters.**

For this problem mark an area of 5 rows and 4 columns. Type in **LINEST(D2:D11, A2:C11, TRUE,TRUE) and press CONTROL+SHIFT+ENTER** to enter this formula into all the marked cells.

Note that the range **D2:D11** is the range where the **dependent variable values** are stored, the range **A2:C11** is the range where the **independent variable** values are stored, the first logical variable **TRUE (**or the number 1) indicates that the parameter $\mathbf{a_0}$ cannot be assumed to be zero and the second logical variable **TRUE** indicates that a matrix of **regression statistics** should also be returned.

It is permitted to mark a one-, two-, three-, four-, or five-row array depending on the amount of information desired.

The results obtained do not include any labeling and **labeling should be added manually**.

Multiple Linear Regression

Results (1)

	a3	a2	a1	a0
coeff.s	4.4446E-05	-75.7482	-9318.66	216.721
std.dev.s	2.0439E-05	23.87706	1984.96	63.921
R^2, SE (y)	0.99975	0.017208	#N/A	#N/A
F, df	8042.39	6	#N/A	#N/A
SS(reg),SS(resid)	7.1446	0.001777	#N/A	#N/A

For obtaining the results reported by POLYMATH **the first three rows are significant**. The first row (coeff.s) contains the values of the parameters. The second row **(std. dev. S**.) contains the **standard deviation** of the parameters. These values can be multiplied by the appropriate value from the t distribution to obtain the 95% **confidence intervals**. The square of the standard error in y (SE y) is the **variance** as reported by POLYMATH.

Multiple Linear Regression

Variance and Confidence intervals and Residuals

Removing the extra rows from the results table and adding the calculations of the confidence intervals and the variance yields the following table (only the first two columns out of the four are shown).

	A	B	C
13		a3	a2
14	coeff.s	=LINEST(D2:D11,A2:C11,1,1)	=LINEST(D2:D11,A2:C11,1,1)
15	std.dev.s	=LINEST(D2:D11,A2:C11,1,1)	=LINEST(D2:D11,A2:C11,1,1)
16	R^2, SE (y)	=LINEST(D2:D11,A2:C11,1,1)	=LINEST(D2:D11,A2:C11,1,1)
17	95% conf. int.	**=B15*2.4469**	**=C15*2.4469**
18	Variance	**=C16^2**	

Note that the *t* value for 95% confidence intervals with 6 degrees if freedom is: $t = 2.4469$.

	E	F
1	logP(calc)	residual
2	**=$E$14+$D$14*A2+$C$14*B2+$B$14*C2**	**=D2-E2**

Polynomial Regression

Options and Instructions

The **LINEST** function and "Regression" tool from the "Analysis ToolPak" can be used for carrying out linear regression. The LINEST function has the advantages over the "Regression" tool that the calculation results are automatically updated when the data is modified and the results are easier to rearrange for documentation purposes. The "Regression" tool provides more statistical data and the output is clearly labeled.

The use of the LINEST function for carrying out polynomial regression will be demonstrated here in reference to Problem 2.3a in the book of Cutlip and Shacham. To prepare the data file arrange the columns of data so that the column of the dependent variable and the column of the independent variable are next to each other and put the column of the independent variable as the last one.

Copy these columns of the data from the POLYMATH data table and paste them into an Excel worksheet. Define additional columns that contain increasing powers of the independent variable, up to the 5^{th} degree.

Polynomial Regression

Excel Formulas and Numerical Values

	A	B	C	D	E	F
1	Cp	TK	TK^2	TK^3	TK^4	TK^5
2	**34.06**	**50**	**=B2^2**	**=B2^3**	**=B2^4**	**=B2^5**
3	41.3	100	=B3^2	=B3^3	=B3^4	=B3^5

Numerical values

	A	B	C	D	E	F
1	Cp	TK	TK^2	TK^3	TK^4	TK^5
2	34.06	50	2.500E+03	1.250E+05	6.250E+06	3.125E+08
3	41.3	100	1.000E+04	1.000E+06	1.000E+08	1.000E+10
4	48.79	150	2.250E+04	3.375E+06	5.063E+08	7.594E+10
5	56.07	200	4.000E+04	8.000E+06	1.600E+09	3.200E+11
6	68.74	273.16	7.462E+04	2.038E+07	5.568E+09	1.521E+12
7	73.6	298.15	8.889E+04	2.650E+07	7.902E+09	2.356E+12
…						
20	**205.89**	**1500**	**2.250E+06**	**3.375E+09**	**5.063E+12**	**7.594E+15**

Polynomial Regression

Using the LINEST Function for a 2nd Degree Polynomial

To solve for a second order polynomial (with three parameters) **mark an area of 3 rows and 3 columns.**

Type in **LINEST(A2:A20, B2:C20, TRUE,TRUE)** and press **CONTROL+SHIFT+ENTER** to enter this formula into all the marked cells.

Note that the range **A2:A20** is the range where the **dependent variable** values are stored, the range **B2:C20** is the range where the **independent variable** values are stored, the first logical variable TRUE (or the number 1) indicates that the parameter a_0 cannot be assumed to be zero and the second logical variable TRUE indicates that a matrix of regression statistics should also be returned.

The results obtained do not include any labeling and **labeling is added manually**.

Polynomial Regression

Results for a 2nd Degree Polynomial

	A	B	C	D
23		a2	a1	a0
24	coeff.s	-6.16E-05	0.21778651	17.7427328
25	std.dev.s	3.53E-06	0.005436794	1.60970868
26	R^2, SE (y)	0.998331	2.610626597	#N/A

Calculation of the confidence intervals and the variance (only the first two columns out of the four are shown).

	A	B	C
23		a2	a1
24	coeff.s	=LINEST(A2:A20,B2:C20,1,1)	=LINEST(A2:A20,B2:C20,1,1)
25	std.dev.s	=LINEST(A2:A20,B2:C20,1,1)	=LINEST(A2:A20,B2:C20,1,1)
26	R^2, SE (y)	=LINEST(A2:A20,B2:C20,1,1)	=LINEST(A2:A20,B2:C20,1,1)
27	95% conf. int.	**=2.1199*B25**	**=2.1199*C25**
28	Variance	**=C26^2**	

Multiple Nonlinear Regression

Instructions

To carry out multiple nonlinear regression an objective function containing the **sum of squares of the errors** is prepared and this objective function is be **minimized by means of the "Solver" tool** by changing the regression model parameters.

Demo 6c is used as an Example.

In this particular example the Antoine equation is fitted to vapor pressure (*Vp*) versus temperature (T °C) data. Thus, the objective function to be minimized is the following.

$$s^2 = \sum_{j=1}^{N} \left[Vp_j - 10\text{^}\left(A + \frac{B}{(C + T_j)} \right) \right]^2 \qquad (5)$$

After copying the independent and dependent variable data from the POLYMATH file and pasting them into an Excel worksheet the objective function can be calculated in three successive columns.

Multiple Nonlinear Regression

Excel Formulas

	A	B	C	D	E
1	A	8.752			
2	B	-2035.33			
3	C	273			
4	Variance	=E18/(10-3)			
5					
6	**TC**	**Pv (mmHg)**	**$(Pv)_{calc}$**	**Residual**	**Sum of Sqrs.**
7			**10^(A+B/(TC+C))**	**$(Pv)-(Pv)_{calc}$**	**of Residuals**
8	-36.7	1	=10^($B$1+$B$2/(A8+$B$3))	=B8-C8	=D8^2
9	-19.6	5	=10^($B$1+$B$2/(A9+$B$3))	=B9-C9	=D9^2

In this table the **initial estimates** for the parameters *A*, *B* and *C* are also shown.

Multiple Nonlinear Regression

Numerical Values at the Initial Estimate

	A	B	C	D	E
4	Variance	6814.45			
5					
6	**TC**	**Pv (mmHg)**	**$\log(Pv)_{calc}$**	**Residual**	**Sum of Squares**
7			**10^(A+B/(TC+C))**	**$(Pv)-(Pv)_{calc}$**	of Residuals
8	-36.7	1	1.3762	-0.3762	0.1415
9	-19.6	5	5.2471	-0.2471	0.0611
10	-11.5	10	9.3050	0.6950	0.4830
11	-2.6	20	16.7841	3.2159	10.3420
12	7.6	40	31.5147	8.4853	71.9996
13	15.4	60	49.5097	10.4903	110.0467
14	26.1	100	88.5439	11.4561	131.2422
…					
17	80.1	760	972.3724	-212.3724	45102.0196
18				**Sum**	**47701.1174**

Multiple Nonlinear Regression

Numerical Values at the Solution

The **sum of squares of errors** is stored in cell **C18**. The **"Solver" tool** is used to minimize this value while changing the values of the parameters **A, B and C (in cells B1, B2 and B3).**

	A	B
1	A	**6.6185**
2	B	**-1054.98**
3	C	**202.14**
4	Variance	**2.2819**